


Podstawy projektowania systemów mikroprocesorowych, część 11



Nawet najprostsze mikrokontrolery '51 można wykorzystać do pomiaru wielkości analogowych (co prawda niezbyt dokładnie) i pomiaru czasu (z tym jest znacznie lepiej). Ponadto, stosunkowo prostymi zabiegami można zwiększyć pojemność pamięci danych i programu. O tym, jak „opanować” te typowe problemy piszemy w artykule.

Komparator analogowy mikrokontrolerów 89Cx051

Dwudziestonóżkowe mikrokontrolery firmy Atmel są wyposażone w wewnętrzny komparator analogowy, którego wejścia są dostępne na wyprowadzeniach linii portu P1.0 („+”) i P1.1 („-”). W celu zapewnienia odpowiednio dużej impedancji wejściowej linie te są pozbawione wewnętrznych rezystorów podciągających. Wejściowy prąd upływu komparatora, według danych producenta, nie przekracza 10 μA , natomiast napięcie niezrównoważenia jest mniejsze od 20 mV. Napięcia

podawane na wyprowadzenia P1.0 i P1.1 muszą mieścić się w przedziale wartości napięcia zasilania (pomiędzy GND a Vcc). Stan wyjścia komparatora (poziom wysoki lub niski) jest zależny od wartości podanego na jego wejście napięcia, i można go odczytać jedynie programowo poprzez odczyt linii portu P3.6 (która nie jest wyprowadzona na zewnątrz). Jedynka odczytana z portu P3.6 oznacza, że napięcie na wyprowadzeniu P1.0 jest większe od napięcia na wyprowadzeniu P1.1. Aby korzystać z komparatora, należy spełnić te same warunki co przy wykorzystywaniu linii portów jako wejść - linie P1.0 i P1.1 muszą być programowo ustawione w stan 1 (o ile ich stan został zmieniony po zerowaniu mikrokontrolera).

Wewnętrzny komparator bywa często wykorzystywany w urządzeniach cyfrowo-analogowych, gdy zachodzi potrzeba określenia wzajemnej zależności (większy-mniejszy) dwóch sygnałów analogowych i przekazania tej informacji do mikrokontrolera. W przypadku, gdy chcemy przekazać do mikrokontrolera informację o wartości napięcia sygnału analogowego, wówczas konieczne jest wykorzystanie przetwornika analogowo-cyfrowego. Jeżeli nie chcemy wykorzystywać żadnego specjalizowanego układu scalonego, to do tego celu również możemy użyć wbudowanego komparatora, stosując układ jak na rys. 35. W tym układzie jest źródło napięcia narastającego quasi liniowo (R i C), które jest doprowadzone do jednego z wyprowadzeń komparatora i porównywane z napięciem wejściowym. Dodatkowym elementem powodującym rozładowywanie kondensatora jest wewnętrzny tranzystor mikrokontrolera zwierający linię portu P1.0 do masy.

Przetwornik działa w następujący sposób: po wpisaniu „1” na linię portu P1.0 tranzystor wyjściowy tego portu zostaje zatkany i rozpoczyna się ładowanie kondensatora C poprzez rezystor R. Równocześnie

List. 15

```
;;Blok inicjalizacji programu głównego
MOV TMOD,#010H           ;Będziemy korzystać z licznika T1 w trybie 1
CLR P1.0                 ;rozładowanie kondensatora

{dalszy ciąg programu}

;;procedura przetwornika A/C
;;zwraca liczbę 16-bitową zawierającą 15-bitowy wynik pomiaru
;;(najstarszy bit równy 0) do zmiennych AC_H i AC_L

PRZETWORNIK:

MOV TL1,#0               ;wyzerowanie
MOV TH1,#0               ;licznika

SETB P1.0                ;rozpoczęcie procesu ładowania C
SETB TR1                 ;i włączenie licznika


JNB P3.6,$               ;czekaj dopóki komparator w stanie niskim

CLR TR1                  ;zatrzymaj licznik
CLR P1.0                 ;rozładuj kondensator


CLR C                    ;wyzeruj znacznik przeniesienia
MOV A,TH1                ;starszy bajt licznika do akumulatora
RLC A                    ;przesuń w lewo przez znacznik przeniesienia
MOV AC_H,A               ;zapamiętaj starsze 7 bitów

MOV A,TL1                ;młodszy bajt licznika do akumulatora
RLC A                    ;przesuń w lewo przez znacznik przeniesienia
MOV AC_L,A               ;zapamiętaj młodsze 8 bitów

RET                       ;powrót do programu głównego
```



Rys. 35



Rys. 36

programowo zostaje uruchomiony wewnętrzny licznik mikrokontrolera. Po czasie potrzebnym na naładowanie kondensatora do wartości napięcia równej napięciu wejściowemu następuje zmiana stanu na wyjściu komparatora, co wykorzystujemy do zatrzymania licznika. Licznik po zatrzymaniu zawiera liczbę zliczonych impulsów proporcjonalną do wartości napięcia U_{we} . Po pomiarze należy rozładować kondensator (wpisanie „0” do P1.0) i można rozpocząć kolejny cykl pomiarowy.

Na list. 15 przedstawiono procedurę realizującą 15-bitowy przetwornik A/C. Na pierwszy rzut oka może bezsensownym wydawać się ignorowanie najmniej znaczącego bitu zawartości licznika (wskutek przesunięcia zliczonej zawartości o 1 bit w lewo), jednak po analizie listingu widać, że ma to sens, po-

nieważ instrukcja sprawdzająca stan komparatora `JNB P3.6,$` jako instrukcja skoku wykonuje się przez dwa cykle maszynowe. Stan komparatora jest więc sprawdzany z częstotliwością o połowę mniejszą niż częstotliwość cykli maszynowych, co powoduje, że najmniejszej wykrywalnej różnicy napięć dwóch kolejnych pomiarów odpowiadać będą stany licznika różniące się właśnie o dwa cykle - najmniej znaczący bit nie będzie się więc zmieniał (tzn. nie da się wykryć zmiany), za każdym razem będzie przyjmował stan 0 lub 1 (ale ciągle ten sam, zależny od kolejności instrukcji w programie przetwornika). Dlatego w celu uzyskania zmiany na najmłodszym bicie należy wynik podzielić przez dwa - co jest równoznaczne z przesunięciem w lewo o 1 bit.

Jeżeli znajdzie potrzeba zwiększenia rozdzielczości przetwornika, to można wykorzystać przerwania licznika i po każdym wywołaniu programu obsługi przerwania zwiększać o jeden zewnętrzną zmienną. W ten sposób można zwiększać rozdzielczość w nieskończoność, jednak praktycznym ograniczeniem jest czas pomiaru oraz dokładność komparatora i użytych elementów. Rozdzielczość możliwa do uzyskania przy użyciu samego timera jest i tak większa niż dokładność wbudowanego komparatora, co powoduje konieczność jego kalibracji oddzielnie dla każdego egzemplarza mikrokontrolera.

Układ o schemacie z rys. 35 działa poprawnie do napięcia wejściowego w zakresie od 0,3 V do połowy napięcia zasilania. Powyżej tej wartości silnie zaznacza się nie-

List. 16

```
;;Odczyt stanów linii układu U4 (adres 0xB000)
MOV DPTR,#0B000H ;wpis adresu do rejestru indeksowego
CLR A ;wyzerowanie akumulatora
MOVC A,@A+DPTR ;odczyt z pamięci programu o adresie z DPTR
; stan linii wejściowych U4 pojawia się w A

;;Pozostałe układy odczytujemy identycznie, wpisując odpowiedni adres do DPTR
```

liniowość charakterystyki ładowania kondensatora ze źródła napięciowego (prąd nie jest stały), natomiast poniżej 0,3 V może być odczuwalny wpływ napięcia odkładającego się na tranzystorze rozładowującym. Dokładność pomiaru można polepszyć stosując ładowanie kondensatora ze źródła prądowego (liniowość charakterystyki ładowania w czasie) oraz dobudowując obwód rozładowania kondensatora z zewnętrznym tranzystorem, najlepiej typu MOSFET o małej rezystancji otwartego kanału (zejście jak najbliższej wartości 0 V).

Wartości elementów R i C (czy źródła prądowego) nie wpływają na dokładność pomiaru, ważne jest jedynie użycie elementów o dobrej stabilności czasowej i temperaturowej w celu uzyskania powtarzalności wyników pomiarów. Jeżeli w układzie jest użyty mikrokontroler 89C5x nieposiadający wbudowanego komparatora, to przetwornik A/C możemy zbudować dodając zewnętrzny komparator. Najlepiej do tego celu nadają się niskonapięciowe (niskie, pojedyncze napięcie zasilania) wzmacniacze operacyjne *rail-to-rail* (ważne ze względu na zakres pomiarowy). Uzyska się wtedy również o wiele lepsze parametry niż przy wykorzystaniu komparatora wbudowanego.

Obsługa pamięci zewnętrznej

Mikrokontrolery 89C5x, podobnie jak 8051, umożliwiają współpracę z zewnętrzną pamięcią danych i programu. Dołączenie zewnętrznej pamięci danych jest stosowane w przypadku programów wykonujących operacje na dużych zbiorach liczb niemieszczących się w pamięci wewnętrznej. Choć dołączanie zewnętrznej pamięci programu wydaje się zbędne ze względu na wbudowaną w mikrokontrolery Atmel (a także w mikrokontrolery innych producentów) pamięć Flash (chyba że okaże się ona za mała), to jednak możliwości, jakie daje utworzenie systemowych szyn danych, adresu i sterującej mogą być z powodzeniem wykorzystane do komunikacji z innymi układami.


Na rys. 36 przedstawiono schemat układu z zewnętrzną pamięcią danych oraz z dodatkowymi portami wejściowymi „udającymi“ zewnętrzną pamięć programu.

Układ działa następująco: mikrokontroler U1 współpracuje ze standardowo dołączonym układem zetrzasku mniej znaczącej połówki adresu U2 (74LS573). Przy dostępie do pamięci zewnętrznej port P0 działa inaczej niż podczas pracy typowej i nie wymaga stosowania rezystorów podciągających. Zatrzaśnięta połówka adresu wraz z doprowadzonymi bezpośrednio z mikrokontrolera siedmioma kolejnymi liniami adresowymi jest doprowadzona do układu pamięci RAM 62256 (32 kB pojemności). Do pamięci doprowadzone są również sygnały sterujące zapisem i odczytem. Układ jest uaktywniany tylko wtedy, gdy na linii adresowej A15 (P2.7) występuje poziom niski (dostęp do pierwszych 32kB pamięci). Wraz z pamięcią do szyny danych mikrokontrolera dołączone są trzy układy wzmacniaczy buforowych (U4, U5, U6 - 74LS541), których wejścia stanowią dodatkowe linie wejściowe systemu. Odczyście takich buforów może być dołączonych więcej lub mniej - ich liczba zależy wyłącznie od potrzeb,

a ograniczeniem górnym jest liczba możliwych do użycia linii adresowych (choć i to ograniczenie można wyeliminować przez dołączenie dodatkowych układów logicznych tworzących dekodery adresu). Odczyt stanu linii na wejściach konkretnego bufora wymaga wykonania odczytu z pamięci programu (aktywacja linii PSEN) spod odpowiedniego adresu (poziom niski pojawiający się na liniach A14, A13 lub A12). Ważne jest, aby podawany adres nie powodował jednoczesnego wyzerowania kilku linii obsługujących bufor, gdyż spowoduje to wystąpienie konfliktu na magistrali. Należy też pamiętać, aby adres ten był wyższy niż maksymalny możliwy adres wewnętrznej pamięci programu, gdyż tylko wtedy układ sterowania mikrokontrolera przełączy się na odczyt pamięci zewnętrznej. Ciąg instrukcji pozwalający na odczyt stanu poszczególnych buforów przedstawiono na list. 16.

Po dodaniu do prezentowanego układu kolejnych zatrzaśków (np. 74LS573) oraz dodatkowego układu logicznego możliwe jest również zbudowanie portu wyjściowego (symulującego pamięć danych). Zastosowanie układu buforów dwukierunkowych umożliwi budowę portu dwukierunkowego.

Należy zwrócić uwagę, że prezentowany sposób, choć powszechnie stosowany w sytuacjach niedoboru portów wyjściowych mikrokontrolera wskutek dołączenia pamięci zewnętrznej, nie ma sensu w przypadku nie instalowania takiej pamięci.



Rys. 37

Wówczas można zastosować rozwiązanie bazujące na standardowej obsłudze portów, np. traktując port P0 jako magistralę danych, a port P1 czy inny jako magistralę adresowo/sterującą. Pozwala to na łatwiejsze i efektywniejsze pisanie programu (eliminacja czasochłonnych instrukcji MOV DPTR,#... czy MOVC lub MOVX). Oszczędza też linie portów mikrokontrolera, gdyż na obsługę takiej realizowanej programowo magistrali systemowej przeznaczamy tyle linii portów, ile naprawdę potrzeba (zamiast dwóch pełnych portów P0 i P2 i kilku linii dodatkowych).

Zliczanie impulsów zewnętrznych przez wewnętrzne liczniki

Wbudowane w mikrokontrolery '51 liczniki T0 i T1 (oraz T2 w wersji '52) zazwyczaj bywają używane jako timery, tzn. zliczają impulsy o częstotliwości cyklu maszynowego mikrokontrolera. Czasami przydatne jest wykorzystanie tych liczników do zliczania impulsów doprowadzonych z zewnątrz do wyprowadzeń mikrokontrolera, oznaczonych jako T0, T1 i T2 (odpowiednio dla kolejnych liczników). Wejścia te stają się źródłem impulsów zliczanych przez liczniki po ustawieniu (poziom wysoki) bitów C/T rejestru TMOD (C/T2 rejestru T2CON dla '52). Zwiększanie zawartości danego licznika odbywa się przy wystąpieniu zbocza opadającego na odpowiadającym mu wejściu. Zliczanie impulsów zewnętrznych ma pewne ograniczenia częstotliwościowe. Stany linii T0, T1 i T2 odczytywane są synchronicznie z cyklem maszynowym mikrokontrolera, a wykrycie zbocza opadającego (zwiększenie zawartości licznika) wymaga wykrycia poziomu wysokiego w jednym cyklu i niskiego w drugim. Stąd maksymalna częstotliwość sygnału o wypełnieniu 50% doprowadzonego do zewnętrznych wejść liczników musi być mniejsza niż 1/24 częstotliwości taktowania procesora (co daje częstotliwość zliczanych impulsów 1 MHz przy zastosowaniu kwarcu 24 MHz). Jeżeli współczynnik wypełnienia byłby różny od 50%, wówczas decyduje czas trwania krótszej fazy - co najmniej 12 taktów zegara (1 cykl maszynowy).

W oparciu o zliczanie impulsów zewnętrznych (w określonej bramce czasowej) działają wszystkie cyfro-

List. 17

```

;;Inicjacja liczników
MOV TMOD,#025H      ;T1 - timer, tryb 2, T0 - licznik, tryb 1
MOV TH1,#6          ;wartość początkowa po przeładowaniu T1
CLR STOP            ;zerowanie bitu informującego o końcu pomiaru
SETB EA             ;włączenie układu przerwań
SETB ET1           ;włączenie przerwania timera T1

CLR P3.3            ;blokada zliczania
CLR P3.2            ;zwolnienie resetu
SETB TR0           ;włączenie zliczania T0 (działa blokada zewnętrzna)

{.....}

;;procedura pomiaru częstotliwości
POMIAR:
MOV DZIEL1,#200     ;inicjacja dodatkowych
MOV DZIEL2,#20      ;dzielników częstotliwości (czas pomiaru 1s)

MOV TH0,#0          ;inicjacja
MOV TL0,#0          ;wartości początkowych
MOV TL1,#6          ;liczników

SETB P3.2           ;zerowanie liczników
CLR P3.2           ;zewnętrznych

CLR STOP           ;wyzerowanie bitu STOP
SETB P3.3          ;włączenie bramki
SETB TR1           ;włączenie licznika T1

JNB STOP,$         ;oczekiwanie na zakończenie pomiaru

CLR TR1            ;wyłączenie licznika T1

MOV WYNIK1,P1       ;zapamiętanie pierwszego bajtu wyniku
MOV WYNIK2,TL0      ;zapamiętanie drugiego bajtu wyniku
MOV WYNIK3,TH0      ;zapamiętanie drugiego bajtu wyniku

RET                ;powrót do programu głównego

;;obsługa przerwania licznika T1
INT_T1:
PUSH PSW           ;zachowanie PSW na stosie
DJNZ DZIEL1,DALEJ  ;zmniejszenie zawartości dzielnika
;i skok jeżeli nie zero
MOV DZIEL1,#200    ;inicjacja zawartości dzielnika
DJNZ DZIEL2,DALEJ ;zmniejszenie zawartości drugiego dzielnika
;i skok jeżeli nie zero

CLR P3.3           ;wyłączenie bramki
SETB STOP          ;ustawienie bitu STOP (minęła 1 sekunda)

DALEJ1:
POP PSW            ;pobranie PSW ze stosu
RETI               ;powrót do programu głównego

```

we częstościomierze. Nic zatem nie stoi na przeszkodzie, aby taki częstościomierz zbudować w oparciu o mikrokontroler '51.

Schemat jednej z najprostszej realizacji takiego przyrządu mierzącego częstotliwość przedstawiono na **rys. 37**. Zawiera jedynie elementy związane z pomiarem częstotliwości - aby uzyskać gotowy przyrząd, należałoby dołączyć przynajmniej układ wyświetlacza lub komunikacji z PC i odpowiednio go oprogramować.

Głównym elementem częstościomierza jest oczywiście mikrokontroler, który realizuje program sterujący pomiarem oraz wykorzystuje wbudowany licznik T0 do zliczania impulsów zewnętrznych. W celu zwiększenia maksymalnej wartości mierzonej częstotliwości, przed wejściem licznika T0 włączono 8-bitowy licznik binarny zbudowany z dwóch układów 7493. Wyjścia tego licznika dołączone są do portu

P1, co umożliwia odczyt ich zawartości. Przed wejściem licznika 7493 włączono wzmacniacz zbudowany na tranzystorze bipolarnym, co pozwala na uzyskanie czułości rzędu kilkuset mV. Dioda D1 zabezpiecza przed ujemnymi półówkami napięcia doprowadzonego sygnału. Sygnał z tranzystora jest z sygnałem z linii P3.3 połączony galwanicznie (iloczyn „na drucie“), co umożliwia bramkowanie wejścia licznika. Mikrokontroler generuje sygnał zerowania zewnętrznych liczników na linii P3.2. Na **list. 17** przedstawiono program realizujący pomiar częstotliwości z wykorzystaniem licznika T1 jako generatora podstawy czasu (bramki czasowej). Ustawienia rejestrów licznika obliczono dla częstotliwości kwarcu 12 MHz.

Program działa następująco: po inicjacji liczników mikrokontrolera wejście do procedury pomiaru częstotliwości powoduje wyzerowanie licznika impulsów zewnętrznych

(T0), zainicjowanie wartości początkowej licznika (timera) podstawy czasu (bramki czasowej T1) oraz współpracujących z nim komórek pamięci wykorzystywanych do przechowywania stanu dodatkowych dzielników koniecznych do uzyskania czasu bramkowania równego jednej sekundzie (ze względu na zbyt małą pojemność licznika). Wykonywane jest również zerowanie liczników zewnętrznych (impulsem wygenerowanym na P3.2). Po tym następuje odblokowanie bramki (linia P3.3) i włączenie odmierzenia czasu impulsu bramkującego (włączenie timera T1). Pomiar przeprowadzany jest do czasu, aż poprzez procedurę obsługi przerwania zostanie ustawiony bit STOP. Wówczas następuje zablokowanie bramki i licznika czasu bramkującego oraz zapamiętanie wyniku pomiaru zarówno z liczników zewnętrznych, jak i licznika wewnętrznego.

Pomiar trwa jedną sekundę. Maksymalna mierzona częstotliwość zależy od szybkości zastosowanych

liczników zewnętrznych (podział przez 256 umożliwia prawidłową pracę licznika wewnętrznego do częstotliwości wejściowych ponad 100 MHz), z tym że w przedstawionym programie częstotliwość mierzona jest w 24-bitowym liczniku (16 bitów licznika wewnętrznego i 8 zewnętrznego), co pozwala na przedstawienie z rozdzielczością 1 Hz maksymalnej częstotliwości równej 16777215 Hz (ponad 16 MHz). W celu zwiększenia długości licznika należałoby wykorzystać flagę przepełnienia licznika TF0 (lub procedurę obsługi przerwania) i zrealizować dodatkowy licznik programowy - dodanie zmiennej 8-bitowej utworzyłoby licznik 32-bitowy, którego pojemność przekracza już możliwości sprzętu (taki licznik przepełniałby się w czasie 1 s przy częstotliwości ponad 4 GHz). Przedstawiona procedura pomiaru nie uwzględnia błędów związanych ze skończonym czasem wykonywania się instrukcji w momentach uruchamiania i zatrzymywania pro-

cedury pomiarowej - bezwzględna wartość błędu wzrasta wraz z wartością mierzonej częstotliwości, co w przypadku dokładnych pomiarów powinno być uwzględnione w postaci poprawki wprowadzanej do wyniku.

Możliwe jest też wykorzystanie wyprowadzenia P3.3/INT1 jako sterującego pracą licznika T1 (ustawienie bitu GATE w rejestrze TMOD dla licznika T1) i dołączenie go do linii sterującej sygnałem bramkowania (konieczna zmiana funkcji wyprowadzeń) - powoduje to jednak zwiększenie zapotrzebowania układu o jedno wyprowadzenie mikrokontrolera, co może nie być bez znaczenia w przypadku „małych“ Atmeli. W takiej konfiguracji bit TR1 ustawiamy na stałe na „1“, a licznik T0 jest synchronizowany z bramki czasowej, co zmniejsza błędy związane z sekwencyjnym, a nie jednoczesnym wykonaniem instrukcji włączenia licznika i włączenia bramki.

Paweł Hadam, EP
pawel.hadam@ep.com.pl